1. Introducción a Mathematica

1.1 Objetivos

1. Conocer la sintaxis básica del lenguaje de Wolfram.
2. Familiarizarse con el programa Mathematica.
3. Realizar sumas de Riemann mediante la utilización del programa Mathematica.
4. Determinar fórmula para el cálculo de integrales mediante sumas de Riemann con la utilización del programa Mathematica.

1.2 Instrucciones

1. Ingresar al enlace `https://goo.gl/r7TVbw`.
2. Replicar en un archivo de Mathematica (llamarlo `ApellidoNombre_Lab01_1.nb`) las secciones: Ingreso de entradas, Fracciones y decimales, Variables y funciones, Álgebra, Gráficos en 2D, Límites, Derivadas e Integrales. Insertar el nombre de sección al inicio de cada una de ellas.
3. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab01_2.nb`), realizar el siguiente procedimiento.

   a) Definir $a = 0$, $b = 1$ y $n = 10$, con esto, definir $\Delta x = \frac{b-a}{n}$.

   ```
   In[1]:= a=0; b=1; n=10;
   In[2]:= Δx=(b-a)/n;
   ```

   b) Realizar una partición de orden 10 del intervalo $[a, b]$. Para esto, utilizar la función `Table`. Además, definir por Cd las etiquetas derechas, por Ci las etiquetas izquierdas y por Cc las etiquetas centrales.

   Para la partición tenemos:

   ```
   In[3]:= P=Table[a+k*Δx,{k,0,n}]
   ```
   Out[3]= $\{0,\frac{1}{10},\frac{1}{5},\frac{3}{10},\frac{2}{5},\frac{1}{2},\frac{3}{5},\frac{7}{10},\frac{4}{5},\frac{9}{10},1\}$

   Para las etiquetas derechas tenemos:

   ```
   In[4]:= Ci=Table[a+(k-1)*Δx,{k,1,n}]
   ```
   Out[4]= $\{0,\frac{1}{10},\frac{1}{5},\frac{3}{10},\frac{2}{5},\frac{1}{2},\frac{3}{5},\frac{7}{10},\frac{4}{5},\frac{9}{10}\}$

   Para las etiquetas izquierdas tenemos:

   ```
   In[5]:= Cd=Table[a+k*Δx,{k,1,n}]
   ```
   Out[5]= $\{\frac{1}{10},\frac{1}{5},\frac{3}{10},\frac{2}{5},\frac{1}{2},\frac{3}{5},\frac{7}{10},\frac{4}{5},\frac{9}{10},1\}$

Para las etiquetas centrales tenemos:

```
In[6]:= Cc=Table[a+(k-1/2)*Δx,{k,1,n}]
```

Out[6]= $\left\{\frac{1}{20},\frac{3}{20},\frac{1}{4},\frac{7}{20},\frac{9}{20},\frac{11}{20},\frac{13}{20},\frac{3}{4},\frac{17}{20},\frac{19}{20}\right\}$

c) Definir la función

$$\begin{aligned} f\colon [0,1] &\longrightarrow \mathbb{R} \\ x &\longmapsto x^3 \end{aligned}$$

y realizar la suma de Riemann de f con la partición y etiquetas del literal anterior.

Para definir la función:

```
In[7]:= f[x_]=x^3;
```

Para la suma de Riemann de etiquetas izquierdas:

```
In[8]:= Si=Sum[f[a+(k-1)*Δx]*Δx,{k,1,n}]
```

Out[8]= $\frac{81}{400}$

Para la suma de Riemann de etiquetas derechas:

```
In[9]:= Sd=Sum[f[a+k*Δx]*Δx,{k,1,n}]
```

Out[9]= $\frac{121}{400}$

Para la suma de Riemann de etiquetas centrales:

```
In[10]:= Sc=Sum[f[a+(k-1/2)*Δx]*Δx,{k,1,n}]
```

Out[10]= $\frac{199}{800}$

d) Utilizando el literal anterior, determinar una fórmula para la suma de Riemann de la función f para una partición regular de orden $N \in \mathbb{N}^*$, con etiquetas derechas $Sd(N)$, etiquetas izquierdas $Si(N)$ y etiquetas centrales $Sc(N)$.

Iniciemos limpiando las variables y definiendo Δx:

```
In[11]:= Clear[Δx,Si,Sd,Sc]; Δx[N_]=(b-a)/N;
```

Para la suma de Riemann de etiquetas izquierdas:

```
In[12]:= Si[N_]=Sum[f[a+(k-1)*Δx[N]]*Δx[N],{k,1,N}]
```

Out[12]= $\frac{1-2\ N+N^2}{4\ N^2}$

Para la suma de Riemann de etiquetas derechas:

```
In[13]:= Sd[N_]=Sum[f[a+k*Δx[N]]*Δx[N],{k,1,N}]
```

Out[13]= $\frac{(1+N)^2}{4\ N^2}$

Para la suma de Riemann de etiquetas centrales:

```
In[14]:= Sc[N_]=Sum[f[a+(k-1/2)*Δx[N]]*Δx[N],{k,1,N}]
```

Out[14]= $\frac{-1+2\ N^2}{8\ N^2}$

e) Determinar el valor de $\displaystyle\int_0^1 x^3\,dx$, determinando el límite cuando N tiene a más infinito de las tres funciones anteriores.

Para las etiquetas izquierdas:

```
In[15]:= Limit[Si[x],x →∞]
Out[15]= 1/4
```

Para las etiquetas derechas:

```
In[16]:= Limit[Sd[x],x →∞]
Out[16]= 1/4
```

Para las etiquetas derechas:

```
In[17]:= Limit[Sc[x],x →∞]
Out[17]= 1/4
```

f) En un solo gráfico, dibujar las funciones Sd, Si y Sc, en función de N, con $N = 2, \ldots 100$, ¿cuál de las tres sumas es "mejor"?

Para la gráfico, tenemos:

```
In[20]:= Plot[{Si[x],Sd[x],Sc[x]},{x,2,100},PlotLegends→"Expressions"]
```

Out[20]=



g) Utilizar el proceso realizado para calcular la suma de Riemann de la función

$$\begin{aligned} f\colon [a,b] &\longrightarrow \mathbb{R} \\ x &\longmapsto x^3, \end{aligned}$$

con $a, b \in \mathbb{R}$ y $a < b$, con para una partición regular de orden $N \in \mathbb{N}^*$, con etiquetas derechas. Luego, determinar el valor de $\int_a^b x^3\,dx$.

Empecemos limpiando las variable y definiendo Δx

```
In[21]:= Clear[a,b,Δx,S];Δx[N_]=(b-a)/N;
```

Tenemos la suma de Riemann:

```
In[22]:= S[N_]=Sum[f[a+(k-1)*Δx[N]]*Δx[N],{k,1,N}]
```

Out[22]= $\dfrac{-a^4+2\ a^3\ b-2\ a\ \ b^3+b^4-2\ a^4\ N+2\ a^3\ b\ N+2\ a\ b^3\ N-2\ b^4\ N-a^4\ N^2+b^4\ N^2}{4\ N^2}$

Tomando el límite, tenemos el valor de la integral:

```
In[23]:= Limit[S[x],x →∞]
```

Out[23]= $\frac{1}{4}\ (-a^4+b^4)$

h) Utilizar el proceso realizado para calcular la suma de Riemann de la función

$$\begin{aligned} f\colon [a,b] &\longrightarrow \mathbb{R} \\ x &\longmapsto e^{2x}, \end{aligned}$$

con $a, b \in \mathbb{R}$ y $a < b$, con para una partición regular de orden $N \in \mathbb{N}^*$, con etiquetas derechas. Luego, determinar el valor de $\int_a^b e^{2x}\, dx$.

Empecemos limpiando las variable y definiendo Δx y $f(x)$:

```
In[33]:= Clear[a,b,Δx,S];Δx[N_]=(b-a)/N;
         f[x_]=Exp[2*x];
```

Tenemos la suma de Riemann:

```
In[35]:= S[N_]=Sum[f[a+(k-1)*Δx[N]]*Δx[N],{k,1,N}]
```

$$\text{Out[35]=} -\frac{(a-b)\ e^{\frac{2a}{N}}\ (e^{2a}-e^{2b})}{(e^{\frac{2a}{N}}-e^{\frac{2b}{N}})\ N}$$

Tomando el límite, tenemos el valor de la integral:

```
In[36]:= Limit[S[x],x →∞]
```

$$\text{Out[36]=} \frac{1}{2}\ (-e^{2a}+e^{2b})$$

4. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab01_3.nb`), realizar el siguiente procedimiento.

   a) Definir a, b números reales tales que $a < b$ y $f : [a, b] \to \mathbb{R}$ una función cualquiera.

   ```
   In[37]:= a=0; b=3; f[x_]=Sin[x];
   ```

   b) Definir funciones tales que $x(k, n)$ represente el elemento de la partición regular de orden $n \in \mathbb{N}^*$ del intervalo $[a, b]$ y $c(k, n)$ represente las etiquetas derechas.

   ```
   In[38]:= x[k_,n_]=a+k*(b-a)/n;
   ```

   ```
   In[39]:= c[k_,n_]=a+(k-1)*(b-a)/n;
   ```

   c) Definir $n = 10$.

   ```
   In[40]:=   n=100;
   ```

   d) Graficar la función f. Utilizar la opción `Prolog` del comando `Plot` para visualizar la suma de Riemann de la función, para esto, utilice la sintexis
   `Prolog -> {Color, Table[Polygon[{Puntos}],{k, 1, n}]}`.

   ```
   In[41]:= Plot[{f[y]},{y,a,b},
                 Prolog →{Cyan,
                     Table[
                         Polygon[{{x[k-1,n],0},{x[k,n],0},{x[k,n],f[c[k,n]]},
                             {x[k-1,n],f[c[k,n]]}}],
                         {k,1,n}
                     ]
                 }
            ]
   ```

Out[41]=



e) Con la ayuda de la función `Manipulate` generar un modelo interactivo para visualizar las sumas de Riemann variando el valor de n.

In[36]:=
```
Manipulate[
    Plot[{f[y]},{y,a,b},
        Prolog →{Blue,
            Table[
                Polygon[{{x[k-1,n],0},{x[k,n],0},
                    {x[k,n],f[c[k,n]]},{x[k-1,n],f[c[k,n]]}}
                ],
                {k,1,n}
            ]
        }
    ],
{n,1,100,1}]
```

Out[36]=



1. INTEGRACIÓN NUMÉRICA

1.1 Objetivos

1. Motivar el uso de integración numérica.
2. Implementar la integración numérica en el programa Mathematica.
3. Estimar los errores de la integración numérica.
4. Comparar los métodos de integración numérica.

1.2 Instrucciones

1. El objetivo de este punto es motivar la integración numérica. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab02_1.nb`), realizar el siguiente procedimiento.

   a) Utilizar el proceso realizado en el laboratorio anterior para calcular la suma de Riemann de la función

$$\begin{aligned} f\colon\ [a,b] &\longrightarrow \mathbb{R} \\ x &\longmapsto e^{x^2}, \end{aligned}$$

   con $a, b \in \mathbb{R}$ y $a < b$, para una partición regular de orden $N \in \mathbb{N}^*$, con etiquetas derechas. Luego, determinar el valor de $\displaystyle\int_1^2 e^{x^2}\,dx$. Explicar por qué el resultado obtenido no es coherente.

   Empecemos las variables básicas:

```
In[1]:= a=1;
        b=2;
        Δx[n_]=(b-a)/n;
        f[x_]=Exp[x^2];
```

   Ahora, definamos la suma de Riemann con etiquetas derechas:

```
In[2]:= S[n_]=Sum[f[a+k*Δx[n]]*Δx[n],{k,1,n}]
```

$$\text{Out[2]=}\ \sum_{k=1}^{N} \frac{e^{\left(1+\frac{k}{N}\right)^2}}{N}$$

   Finalmente, tomemos el límite para obtener el valor de la integral:

```
In[3]:= Limit[S[n],n→∞]
Out[3]= 0
```

   Este resultado no es coherente dado que se está integrando una función continua y estricta positiva.

   b) Utilizar una tabla para calcular la suma de Riemann de orden 10, 20, ..., 100 (para una mejor visualización de la tabla, utilizar el comando `TableForm` con la opción `TableHeadings -> {None, {"n", "S[n]"}}`)
   ¿A qué valor parece que se aproxima la integral?

```
In[4]:= TableForm[
            Table[{k,N[S[k]]},{k,10,100,10}],
            TableHeadings→{None,{"n","S[n]"}}]
```

Out[4]//TableForm=

n	S[n]
10	17.7608
20	16.3313
30	15.8744
40	15.6496
50	15.5159
60	15.4272
70	15.3642
80	15.317
90	15.2804
100	15.2511

Al parecer, la integral se aproxima a 15,2.

c) Calcular la suma con una partición de orden 10000.

```
In[5]:= N[S[10000]]

Out[5]= 14.9926
```

d) Calcular la suma con una partición de orden 1000000.

```
In[6]:= N[S[1000000]]

Out[6]= 14.99
```

2. El objetivo de este punto es implementar y comparar los métodos de integración numérica. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab02_2.nb`), realizar el siguiente procedimiento.

   a) Definir a, b números reales tales que $a < b$ y $f\colon [a,b] \to \mathbb{R}$ una función cualquiera. Además, definir el Δx en función de n.

   ```
   In[1]:= a=1;
           b=2;
           f[x_]=Exp[x^2];
           Δx[n_]=(b-a)/n;
   ```

   b) Definir los métodos de sumas de Riemann con etiquetas derechas R, la regla del trapecios T y la regla de Simpson S.

   Para la suma de Riemann

   ```
   In[2]:= R[n_]=Sum[f[a+k*Δx[n]]*Δx[n],{k,1,n}];
   ```

   Para la regla de trapecios:

   ```
   In[3]:= T[n_]=(f[a]+2*Sum[f[a+k*Δx[n]],{k,1,n-1}]+f[b])*Δx[n]/2;
   ```

   Para el método de Simpson

```
In[4]:= S[n_]=(f[a]+2*Sum[f[a+2*k*Δx[n]],{k,1,n/2-1}]
                +4*Sum[f[a+(2*k-1)*Δx[n]],{k,1,n/2}]+f[b])*Δx[n]/3;
```

c) Utilizar una tabla para calcular la integral aproximada de orden 10, 20, ..., 100 para cada método. Para una mejor visualización, utilizar el comando `TableForm` y la opción `TableHeadings`.

```
In[5]:= TableForm[Table[{k,N[R[k]],N[T[k]],N[S[k]]},{k,10,100,10}],
          TableHeadings →{None,{"n","R[n]","T[n]","S[n]"}}]
```

Out[5]//TableForm=

n	R[n]	T[n]	S[n]
10	17.7608	15.1668	14.9925
20	16.3313	15.0343	14.9901
30	15.8744	15.0097	14.99
40	15.6496	15.0011	14.99
50	15.5159	14.9971	14.99
60	15.4272	14.9949	14.99
70	15.3642	14.9936	14.99
80	15.317	14.9927	14.99
90	15.2804	14.9922	14.99
100	15.2511	14.9918	14.99

d) Para $n \in \mathbb{N}^*$, generar tablas que guarden los datos de la integración numérica de cada método con orden par menor que n.

```
In[6]:= n=100;
        LR=Table[N[R[k]],{k,2,n,2}];
        LT=Table[N[T[k]],{k,2,n,2}];
        LS=Table[N[S[k]],{k,2,n,2}];
```

e) En un solo gráfico, dibujar las tres listas anteriores (utilizar el comando `ListLinePlot`), ¿cuál de los tres métodos es "mejor"? Describa rápidamente el gráfico que se obtiene.

```
In[7]:= ListLinePlot[{LR,LT,LS},PlotLegends→{"LR","LT","LS"}]
```

Out[7]=



f) Presentar gráficos similares a los anteriores para las aproximaciones de:

$$\int_0^{\pi/2} \cos(x)\,dx \qquad \text{y} \qquad \int_0^{2\pi} \cos(x^2)\,dx.$$

Describa rápidamente el gráfico que se obtiene.

El código para realizarlo sería:

In[8]:=
```
Clear[a,b,f,Δx,R,T,S,n,LR,LT,LS]
a=0;
b=10;
f[x_]=Exp[-x^2];
Δx[n_]=(b-a)/n;
R[n_]=Sum[f[a+k*Δx[n]]*Δx[n],{k,1,n}];
T[n_]=(f[a]+2*Sum[f[a+k*Δx[n]],{k,1,n-1}]+f[b])*Δx[n]/2;
S[n_]=(f[a]+2*Sum[f[a+2*k*Δx[n]],{k,1,n/2-1}]
        +4*Sum[f[a+(2*k-1)*Δx[n]],{k,1,n/2}]+f[b])*Δx[n]/3;
n=100;
LR=Table[N[R[k]],{k,2,n,2}];
LT=Table[N[T[k]],{k,2,n,2}];
LS=Table[N[S[k]],{k,2,n,2}];
ListLinePlot[{LR,LT,LS},PlotLegends →{"LR","LT","LS"}]
```

Obteniendo los siguientes gráficos, para $\int_0^{\pi/2} \cos(x)\,dx$:

Out[8]=



Para $\int_0^{2\pi} \cos(x^2)\,dx$:

Out[9]=



3. El objetivo de este punto es analizar los errores para los métodos de integración numérica. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab02_3.nb`), realizar el siguiente procedimiento.

   a) Definir a, b números reales tales que $a < b$ y $f\colon [a,b] \to \mathbb{R}$ una función cualquiera.

   In[1]:=
   ```
   Clear[a,b,f,ET,ES,n,ET,ES]
   a=0;
   b=1;
   f[x_]=Exp[x^2];
   ```

   b) Definir por MT al máximo de $|f''|$, para esto, utilice el comando `NMaximize`.

In[2]:= `MT=NMaximize[{Abs[D[f[x],{x,2}]],a≤x≤b},x][[1]]`

Out[2]= `16.3097`

c) Definir por MS al máximo de $|f^{(4)}|$.

In[3]:= `MS=NMaximize[{Abs[D[f[x],{x,4}]],a≤x≤b},x][[1]]`

Out[3]= `206.589`

d) Definir por $ET(n)$ y $ES(n)$ a la estimación del error de calcular $\int_a^b f(x)\,dx$, por la regla de trapecios y por la regla de Simpson, respectivamente, con orden n.

In[4]:= `ET[n_]=MT*(b-a)^3/(12*n^2)`

Out[4]= $\dfrac{1.35914}{n^2}$

In[5]:= `ES[n_]=MS*(b-a)^5/(180*n^4)`

Out[5]= $\dfrac{1.14772}{n^4}$

e) Graficar las funciones ET y ES. ¿Cuál método es mejor? Describa brevemente el gráfico que obtuvo.

In[6]:= `Plot[{ET[n],ES[n]},{n,2,100},PlotLegends→"Expressions"]`

Out[6]= 

Se tiene una mejor estimación del error para la regla de Simpson.

f) Realizar lo anterior para las integrales

$$\int_0^{\pi} \cos(x)\,dx \qquad \text{y} \qquad \int_0^{2\pi} \cos(x^2)\,dx$$

y comente los resultados.

Bajo una sintaxis similar a la elaborada en la parte anterior, se tiene para $\displaystyle\int_0^{\pi} \cos(x)\,dx$ el gráfico:

Out[7]= 

Para $\displaystyle\int_0^{2\pi} \cos(x^2)\,dx$ el gráfico:

Out[8]=



4. El objetivo de este punto es aplicar los conocimientos de integración numérica a dos ejercicios prácticos.

Ejercicio 1: Si se conoce que la integral

$$\int_{-\infty}^{\infty} f(x)\,dx$$

converge, un método para calcular numéricamente este valor es aproximarlo por

$$\int_{-n}^{n} f(x)\,dx,$$

con $n \in \mathbb{N}^*$, y esta integral calcularla por algún método numérico, pero de orden n^2. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab02_4.nb`), realizar el siguiente procedimiento.

a) En base al método expuesto, indicar por qué no sería buena idea calcular la integral con un método de orden n.

No es buena idea dado que el intervalo sobre el que se integra crecería a la misma velocidad que aumenta el número de partición realizada, con esto, el grosor de la partición no decrecería y no se estaría aproximando el valor de la integral.

b) Sabiendo que

$$\int_{-\infty}^{+\infty} \frac{1}{x^2+1}\,dx = \pi,$$

implementar el método expuesto para aproximar el valor de la integral y así calcular una aproximación de π.

```
In[1]:= Clear[f,n,T,Δx,a,b];
        f[x_]=N[1/(x^2+1)];
        n=1000;
        a=-n;
        b=n;
        Δx=(b-a)/n^2;
        T=N[(f[-n]+2*Sum[f[-n+k*Δx],{k,1,n^2-1}]+f[n])*Δx/2]
Out[1]= 3.13959
```

c) Sabiendo que

$$\int_{0}^{1} \frac{4}{x^2+1}\,dx = \pi,$$

determinar el valor de n para el cual, utilizando la regla de Simpson, se pueda calcular el valor de π con 8 cifras decimales exactas.

```
In[2]:= Clear[f,n,T,Δx,a,b,x,MS];
        a=0;
        b=1;
        f[x_]=4/(1+x^2);
        MS=NMaximize[{Abs[D[f[x],{x,4}]],a≤x≤b},x][[1]]

Out[2]= 96.

In[3]:= Reduce[MS*(b-a)^5/(180*n^4)<0.000000005,n]

Out[3]= n<-101.627||n>101.627
```

Dado que n debe ser para, el valor adecuado para n es 102.

d) Calcule el valor de π con las 8 cifras decimales exactas. Despliegue el resultado con 15 cifras y comente el resultado.

```
In[4]:= n=102;
        Δx=(b-a)/n;
        T=N[(f[a]+2*Sum[f[a+2*k*Δx],{k,1,n/2-1}]
            +4*Sum[f[a+(2*k-1)*Δx],{k,1,n/2}]+f[b])*Δx/3,15]
        N[Pi,15]

Out[4]= 3.14159265358976

Out[5]= 3.14159265358979
```

Como se puede apreciar, se obtienen mucho más que 8 cifras decimales exactas.

1. OBJETIVOS

1. Realizar gráficos en tres dimensiones.
2. Realizar gráficas de campos vectoriales.

2. INSTRUCCIONES

Antes de realizar el procedimiento, empiece manipulando el archivo `EjemplosGraficasVectoriales.nd` para aprender los comandos básicos de graficación.

2.1 Sólidos de revolución

El objetivo de este punto es visualizar sólidos de revolución y calcular su volúmenes y áreas. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab03_1.nb`), realizar el siguiente procedimiento.

Ejercicio 1: Volumen de un mullo

En este ejercicio, visualizaremos el sólido que resulta al realizar un orificio cilíndrico (de radio $3/2$) en una esfera (de radio 3).

1. Defina las funciones
$$\begin{aligned} f\colon [-3,3] &\longrightarrow \mathbb{R} \\ x &\longmapsto \sqrt{9-x^2} \end{aligned} \qquad \text{y} \qquad \begin{aligned} g\colon [-3,3] &\longrightarrow \mathbb{R} \\ x &\longmapsto 3/2. \end{aligned}$$
2. Grafique las funciones f y g, rellenando la región entre las gráficas de estas funciones, utilizar la opción `Filling -> {1 -> {2}}`.
3. Dibuje el sólido generado al rotar las gráficas de las funciones f y g, alrededor del eje x.
4. Determine los puntos donde las dos funciones se cortan.
5. Dibujar la región comprendida por las dos funciones únicamente en los puntos de corte, además, graficar el sólido generado por esta región al gira alrededor del eje x.
6. Para mejorar la visualización, realice únicamente $3/4$ de la rotación, luego, realice la rotación completa, pero únicamente con las funciones restringidas a $[-3, 1{,}5]$.
7. Calcule el volumen del sólido.

Ejercicio 2: Trompeta de Torichelli

En este ejercicio, visualizaremos la denominada Trompeta de Torichelli y calcularemos su volumen y área lateral.

1. Defina la función
$$\begin{aligned} f\colon\ [1,+\infty[&\longrightarrow \mathbb{R} \\ x &\longmapsto 1/x. \end{aligned}$$
2. Grafique la función f y el sólido resultante al girar la gráfica de f al rededor del eje x.
3. Calcule el volumen del sólido generado en el paso anterior.
4. Calcule el área del sólido del paso anterior.
5. Las dos últimas respuestas, ¿son contradictorias?

2.2 Gráfica de funciones vectoriales

El objetivo de este punto es visualizar los diferentes tipos de funciones vectoriales. En un archivo de Mathematica (llamarlo `ApellidoNombre_Lab03_2.nb`), realizar el siguiente procedimiento.

Ejercicio 1: Trayectorias en dos dimensiones

En este ejercicio, generaremos una interactividad para visualizar el trazo de una trayectoria en $\mathbb{R}^2$.

1. Defina la función
$$\begin{aligned} \alpha\colon\ [0,2\pi] &\longrightarrow \mathbb{R}^2 \\ t &\longmapsto (2(1-\cos(t))\cos(t),\ 2(1-\cos(t))\operatorname{sen}(t)). \end{aligned}$$
2. Grafique la trayectoria α en $\mathbb{R}^2$.
3. Con ayuda del comando `Manipulate`, genere una interactividad que vaya graficando la trayectoria.

Ejercicio 2: Trayectorias en tres dimensiones

En este ejercicio, generaremos una interactividad para visualizar el trazo de una trayectoria en $\mathbb{R}^3$.

1. Defina la función
$$\begin{aligned} \alpha\colon\ [0,20\pi] &\longrightarrow \mathbb{R}^3 \\ t &\longmapsto (\cos(t/10)(5+2\cos(t)),\ \cos(t/10)(5+2\cos(t)),\ 2\operatorname{sen}(t)). \end{aligned}$$
2. Grafique la trayectoria α en $\mathbb{R}^3$.
3. Con ayuda del comando `Manipulate`, genere una interactividad que vaya graficando la trayectoria.

Ejercicio 3: Curvas de nivel de campos escalares

En este ejercicio, realizaremos comparaciones entre las curvas de nivel de un campo escalar de $\mathbb{R}^2$ y sus conjuntos de nivel.

1. Defina la función
$$\begin{aligned} f\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R} \\ (x,y) &\longmapsto \cos(10xy). \end{aligned}$$

2. Realice el gráfico de las curvas de nivel de la función f para los valores dentro del cuadrado $[-1,1]^2$. A partir de esto, elabore una descripción de cómo sería la gráfica de la función.

3. Realice el gráfico de la función f y compare con la descripción dada en el paso anterior.

4. Ahora, defina las siguientes funciones
$$\begin{aligned} h\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R} \\ (x,y) &\longmapsto (x+y)^3 \end{aligned} \qquad \text{y} \qquad \begin{aligned} g\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R} \\ (x,y) &\longmapsto (x+y)^5. \end{aligned}$$

5. Realice el gráfico de las curvas de nivel de las funciones h y g, ¿qué nota en común entre estos gráficos?

6. Realice el gráfico de las funciones h y g, utilice la opción `PlotStyle -> Directive[Opacity[0.8]]` para mejorar la visualización de ambas funciones. Compare con la descripción dada en el paso anterior.

Ejercicio 4: Gráfica de superficies

En este ejercicio, utilizaremos lo aprendido hasta este punto, para generar la gráfica de las superficies de ecuaciones
$$z = x^2 + y^2 \qquad \text{y} \qquad (x-1)^2 + y^2 = 1.$$
Con esto, describa la forma de la intersección de estas superficies.

Ejercicio 5: Gráfica de campos vectoriales 2D

En este ejercicio, utilizaremos lo aprendido hasta este punto, para visualizar una relación entre las derivadas parciales de una función y sus curvas de nivel.

1. Defina la función
$$\begin{aligned} f\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R} \\ (x,y) &\longmapsto x^2 - y^2. \end{aligned}$$

2. Defina el campo vectorial
$$\begin{aligned} g\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R}^2 \\ (x,y) &\longmapsto \left(\frac{\partial f}{\partial x}(x,y),\ \frac{\partial f}{\partial y}(x,y) \right). \end{aligned}$$

3. Grafique, en un mismo gráfico, el campo vectorial g y las curvas de nivel de f, para esto, guarde cada gráfico en una variable y luego utilice el comando `Show`.

4. ¿Observa alguna relación entre estos dos grráficos?, pruebe cambiando la función f para mejorar la conjetura que obtenga.

Ejercicio 6: Gráfica de campos vectoriales 3D

En este ejercicio, utilizaremos lo aprendido hasta este punto, para visualizar una relación entre las derivadas parciales de una función y sus superficies de nivel.

1. Defina la función
$$\begin{aligned} f\colon \quad \mathbb{R}^3 &\longrightarrow \mathbb{R} \\ (x,y,z) &\longmapsto x^2+y^2-z^2. \end{aligned}$$

2. Defina el campo vectorial
$$\begin{aligned} g\colon \quad \mathbb{R}^2 &\longrightarrow \mathbb{R}^2 \\ (x,y) &\longmapsto \left(\frac{\partial f}{\partial x}(x,y,z),\ \frac{\partial f}{\partial y}(x,y,z),\ \frac{\partial f}{\partial z}(x,y,z)\right). \end{aligned}$$

3. Grafique, en un mismo gráfico, el campo vectorial g, restringido a una superficie de nivel de f.

4. ¿Observa alguna relación entre estos dos gráficos?, pruebe cambiando la función f o la superficie de nivel para mejorar la conjetura que obtenga.